

Integration eines beliebigen externen Druckservers in APEX

Niels de Bruijn, MT AG

Externe Web-Inhalte können mit Standardbordmitteln von APEX einfach in einer Seite integriert werden. Diese sogenannte „URL-Integration“ wurde bereits in der [Community](#) erläutert. Der Vorteil dieser Konfiguration: der Benutzer merkt gar nicht dass Inhalte von einer externen Quelle gekommen sind. Auch braucht der Benutzer keinen direkten Zugriff auf die externe Quelle.

Aber wie funktioniert dieses Verfahren, wenn die URL anstelle von HTML eine Datei zurückliefert? So kommt es häufig vor, dass ein PDF Dokument über APEX beim Druckserver angefordert und direkt im Browser dargestellt werden soll.




Abb. 1: APEX fordert ein PDF-Dokument beim Druckserver an und gibt es an den Browser zurück.

In diesem Tipp wird beispielhaft die Anbindung von einem "Business Objects"-Druckserver gezeigt. Es ist jedoch möglich, diese Integrationsart auch für andere Reports-Server, wie zum Beispiel Oracle Reports, einzusetzen.

1. Abfrage PDF Dokument testen

Zuerst sollte geprüft werden, ob der Druckserver einwandfrei funktioniert und ob die URL, mit welcher der Bericht angefordert werden soll, korrekt arbeitet. Testen Sie die Berichts-URL daher zunächst im Browser ...

```
http://host:port/CrystalReports/viewrpt.cwr?id=29404&apsuser=apexReporter&apspassword=password&apsauthtype=secEnterprise&promptex-jahr="2009"&promptOnRefresh=0&cmd=export&export_fmt=u2fpdf:0
```

Ein PDF Dokument sollte nun im Browser dargestellt werden. Wenn Sie die Möglichkeit haben, testen Sie diese auch auf dem Server, auf dem die APEX-Datenbank installiert ist. So stellen Sie sicher dass die Datenbank ohne Probleme auf die Webschnittstelle des Business Objects Servers zugreifen kann.

2. UTL_HTTP

Geben Sie das Workspace Schema die Berechtigung das PL/SQL Paket UTL_HTTP auszuführen:

```
sqlplus /nolog
conn sys/password@orcl as sysdba
grant execute on utl_http to my_apex_workspace_schema;
```

Wenn Sie eine 11g-Datenbank betreiben, müssen Sie zusätzlich den Netzwerkzugriff auf den Druckserver explizit genehmigen. Weiteres dazu finden Sie in diesem [Tipp](#).

3. PL/SQL Prozedur für den Aufruf installieren

Die nachfolgende Prozedur ist im Workspace Schema zu installieren:

```
sqlplus /nolog
conn my_apex_workspace_schema/passwort@orcl
set define off

create or replace procedure p_bericht_generieren
is

    l_url          varchar2(32767 char);
    l_http_request  utl_http.req;
    l_http_response utl_http.resp;
    l_blob          blob;
    l_raw           raw(32767);

begin
    -- Aufruf des Druckservers (in diesem Fall Business Objects)
    l_url:= 'http://host:port/CrystalReports/viewrpt.cwr' ||
            '?id=29404' ||
            '&apsuser=apexReporter' ||
            '&apspassword=passwort' ||
            '&apsauthtype=secEnterprise' ||
            '&promptex-jahr="' || v('P1_JAHR') || '"' ||
            '&promptOnRefresh=0' ||
            '&cmd=export' ||
            '&export_fmt=u2fpdf:0'
            ;

    -- HTTP Request
    utl_http.set_transfer_timeout(30);
    l_http_request:= utl_http.begin_request(l_url);
    utl_http.set_header(l_http_request, 'User-Agent', 'Mozilla/4.0');

    -- HTTP Response (PDF document) wird in einem temporären BLOB geladen
    l_http_response:= utl_http.get_response(l_http_request);
    dbms_lob.createtemporary(l_blob, false);
    begin
        loop
            utl_http.read_raw(l_http_response, l_raw, 32767);
            dbms_lob.writeappend(l_blob, utl_raw.length(l_raw), l_raw);
        end loop;
    exception
        when utl_http.end_of_body -- end of response
        then
            utl_http.end_response(l_http_response);
    end;

    -- Der BLOB wird an den Browser versendet
    owa_util.mime_header('application/pdf', false);
    http.p('Content-length: ' || dbms_lob.getlength(l_blob));
    http.p('Content-Disposition: attachment;filename="MyReport.pdf"');
    owa_util.http_header_close;
    wpg_docload.download_file(l_blob);
    dbms_lob.freetemporary(l_blob);

exception
    when utl_http.request_failed
    then
```

```
http.p('Der Druckserver reagiert nicht.');
```

```
when others
```

```
then
```

```
    http.p(sqlerrm);
```

```
end p_bericht_generieren;
```

```
/
```

4. Seite für die Eingabe der Berichtsparemeter erstellen

Gehen Sie wie folgt vor:

- Eine neue **Anwendung** mit einer **leeren Seite** erstellen
- Ein **Region** vom Typ HTML ist mit der Erstellung der Anwendung bereits auf der Seite vorhanden
- Eine **Schaltfläche** in einer Regionsposition erstellen und mit dem Label „Bericht anzeigen“ versehen. Die auszuführende Aktion ist „Seite weiterleiten und an URL umleiten“.

Pro Parameter ein Element erstellen. In diesem Beispiel erstellen wir den Parameter „Jahr“:

- Ein **Element** „P1_JAHR“ erstellen und die folgenden Eigenschaften setzen:
 - Anzeigen als: Textfeld
 - Label: Jahr
 - Verwendete Quelle: Vorhandene Werte im Session-Zustand immer ersetzen
 - Quelltyp: Element (Anwendungs- oder Seitenelementname)
 - Quellwert: P1_JAHR
- Eine **Verzweigung zu Seite** erstellen und die folgenden Eigenschaften überprüfen:
 - Verzweigungspunkt: Nach Verarbeitung (Nach Berechnung, Validierung und Verarbeitung)
 - Zieltyp: Seite in dieser Anwendung
 - Seite: 0
 - Anfordern: APPLICATION_PROCESS=bericht_generieren

5. Ein Anwendungsprozess erstellen (siehe „Gemeinsame Komponenten“)

Die folgenden Eigenschaften sind dabei zu prüfen:

- Name: bericht_generieren
- Ausführungspunkt: Bedarfsgesteuert: Führt den Anwendungsprozess aus, wenn dies von einem Seitenprozess angefordert wird.
- Prozesstext: **p_bericht_generieren;**

Führen Sie jetzt die Anwendung aus, tragen Sie einen Wert für den Parameter ein und klicken Sie auf „Bericht anzeigen“. Obwohl Sie sich immer noch im Kontext von APEX befinden, wird das PDF Dokument nun zum Download angeboten.

Über den Autor und die MT AG:

[Niels de Bruijn](#) arbeitet seit 2003 als Senior Systemberater für die MT AG in Ratingen und hat langjährige Projekterfahrung mit dem Produkt Oracle Application Express.

Die [MT AG](#) ist langjähriger IT-Beratungspartner von Großunternehmen und Mittelstand. Sie steht für Technologie-Know How und praxisnahe, effiziente IT-Dienstleistung: von Strategie und Beratung über Entwicklung und Integration bis hin zu Wartung und Administration von IT-Infrastrukturen.

Und jetzt noch die kleinen Buchstaben, zusammen „Haftungshinweise“ genannt:

In keinem Fall haftet die MT AG für irgendwelche direkten, indirekten, speziellen oder sonstigen Folgeschäden, die sich aus der Nutzung dieses Dokument ergeben. Ausgeschlossen ist auch jegliche Haftung für entgangenen Gewinn, Betriebsunterbrechung, Verlust von Programmen oder sonstigen Daten. Dies gilt auch dann, wenn ausdrücklich auf die Möglichkeit solcher Schäden hingewiesen wird.